

Capitolul 6 – partea a II-a

Tehnici de programare în limbaj de asamblare

6.7 Proceduri recursive

În limbajul de asamblare nu există restricții asupra recursivității: orice procedură proc_a se poate apela pe ea însăși (recursivitate directă) sau poate apela o procedură proc_b care, la rândul ei apelează procedura proc_a (recursivitate indirectă).

În dezvoltarea unei proceduri recursive trebuie pornit de la specificarea algoritmului recursiv. Specificarea unui asemenea algoritm pune în evidență un caz de bază (care oprește recursivitatea) și un caz general, în care se invocă același algoritm, aplicat altui set de date. Specificarea corectă a cazului de bază și garantarea faptului că acesta este atins întotdeauna, indiferent de setul de date de intrare primit, sunt punctele-cheie ale proiectării unui algoritm recursiv.

Să considerăm, ca exemplu, o procedură de afișare în baza 10 a unui număr pe 16 biți fără semn. La nivelul funcțiilor de intrare/iesire, avem posibilitatea de a afișa caractere ASCII. Dorim deci un algoritm recursiv care să genereze cifrele zecimale ale numărului, în ordinea afișării. În §2.2.5 a fost dat un algoritm nerecursiv, care producea cifrele zecimale ale numărului în ordine inversă.

Algoritmul recursiv se specifică în felul următor:

```
putu_proc (n)
{
    dacă (n < 10)
        afișează (n + '0');
    altfel {
        putu_proc (n/10);
        afișează (n MOD 10 + '0');
    }
}
```

Cazul de bază este ($n < 10$), pentru care se face doar afișarea cifrei și revenirea în programul apelant. În Figura 6.7 sunt ilustrate apelurile recursive în cazul afișării numărului 123 și valorile succesive ale parametrului n.

Regulile generale de implementare a procedurilor recursive sunt:

- fiecare apel al procedurii nu trebuie să afecteze datele apelurilor precedente (procedura să fie reentrantă);
- nu se folosesc variabile locale alocate static, ci numai în stivă sau în registre;
- parametrii se transmit prin stivă (se asigură implicit faptul că parametrii sunt locali fiecărui apel);
- registrele ale căror valori au semnificație de la un apel la altul se salvează în stivă;
- dacă procedura întoarce valori prin registre, rezultatele intermediare trebuie memorate în stivă.




Figura 6.7 Apelul recursiv al procedurii putu_proc

Pentru implementarea procedurii `putu_proc`, se va utiliza funcția DOS 2, care afișează la consolă caracterul primit în registrul DL. Pentru accesul la stivă, considerăm structura şablon:

```

sablon struc
    _bp      dw ?
    _cs_ip   dw ?
    n        dw ?
sablon ends
  
```

Pentru o implementare mai eficientă, rescriem algoritmul recursiv în forma:

```

putu_proc(n)
{
    dacă (n < 10)
        x = n;
    altfel {
        putu_proc(n/10);
        x = n MOD 10;
    }
    afișează (x + '0');
}
  
```

Alocarea variabilelor este următoarea:

- `n` - în stivă, accesibil prin expresia `[bp].n`
- `x` - în registrul DL (care se salvează/restaurează)

Implementarea este următoarea:

```

putu_proc proc far
    push  bp          ; Secvența
    mov   bp,sp       ; standard
    push  dx          ; Salvări
    push  ax          ; registre
    push  bx          ; folosite
;
    mov   ax,[bp].n    ; Test caz
    cmp   ax,10        ; de bază
    mov   dl,al        ; x <- n
    jb    p_u_1        ; Salt la afișare
;
    mov   bx,10        ; Caz general
    xor   dx,dx       ; Se calculează
  
```

```

        div    bx          ; n/10 și n MOD 10
                ; AX = n/10
                ; DX (DL) = x = n MOD 10
        push   ax          ; Apel recursiv cu
        call   far ptr putu_proc ; n/10 ca parametru
p_u_1:
        add    dl, '0'    ; x + '0'
        mov    ah, 2      ; Cod funcție DOS
        int    21H        ; Afisare
;
        pop    bx          ; Refaceri
        pop    ax          ; registre
        pop    dx
        pop    bp
        retf   2           ; Revenire cu descărcarea stivei
putu_proc endp

```

În implementarea de mai sus, este esențială integritatea variabilei x de la un apel la altul, adică salvarea/restaurarea registrului DX. Pentru a preveni depășirile, se face împărțire 32 de biți la 16 biți, deci constanta 10 se încarcă într-un registru de 16 biți. După împărțire, DX (de fapt DL) conține restul n MOD 10, iar AX câtul (n/10). Se pune deci AX în stivă și se apelează recursiv procedura. După revenire, se afișează cifra din DL.

Putem acum dezvolta o procedură care să afișeze un număr pe 16 biți cu semn, după algoritmul:

```

puti_proc (n) {
    dacă (n < 0) {
        n = - n;
        afișează ('-');
    }
    putu_proc (n)
}

```

Dacă n e negativ, se complementează față de 2 și se afișează semnul '-', după care se afișează numărul n complementat. Implementarea este următoarea:

```

puti_proc proc far
    push   bp          ; Secvența
    mov    bp, sp       ; standard
    push   ax          ; Salvări
    push   dx          ; registre
;
    mov    ax, [bp].n   ; Parametrul n
    or     ax, ax       ; Poziționare bistabili
    jns   p_i_1         ; Test bit de semn
    mov    dl, '-'
    mov    ah, 2         ; Afisare
    int    21H          ; semn '-'
    neg    word ptr [bp].n ; Complementare n
p_i_1:
    push [bp].n         ; Pregătire parametru
    call  far ptr putu_proc ; Apel putu_proc
;
    pop    dx          ; Refaceri
    pop    ax          ; registre
    pop    bp
    retf   2           ; Revenire cu descărcarea stivei
puti_proc endp

```

Se observă plasarea forma directă de plasare a parametrului [bp].n în stivă, pentru apelul procedurii putu_proc.

Putem scrie acum și macroinstructiuni de apel adecvate:

```
puti macro X
    push  X
    call   far ptr puti_proc
endm
putu macro X
    push  X
    call   far ptr putu_proc
endm
```

După modelul procedurii putu_proc, putem scrie o procedură care să afișeze la consolă reprezentarea unui număr pe 16 biți fără semn într-o bază de numerație dată. Baza se consideră în domeniul 2...36, iar cifrele într-o bază mai mare ca 10 sunt '0'...'9', 'A', 'B' etc. Limitarea bazei la valoarea 36 derivă din această alegere a cifrelor.

Algoritmul este următorul:

```
put_base (n, baza) {
    dacă (baza < 2 sau baza > 36)
        return;
    dacă (n < baza)
        x = n;
    altfel {
        put_base (n/baza, baza);
        x = n MOD baza;
    }
    dacă (x < 10)
        afișează (x + '0');
    altfel
        afișează (x + 'A' - 10);
}
```

Se începe printr-un test de corectitudine a bazei: dacă aceasta nu este în domeniul admisibil, se revine imediat în programul apelant. Se execută apoi același algoritm recursiv ca în procedura putu_proc, înlocuind constanta 10 cu variabila baza. Afisarea cifrei curente x se face diferențiat: dacă este mai mică sau egală decât 10, se afișează ca cifra zecimală; altfel, se afișează ca literă.

Pentru accesul la stivă, se utilizează structura şablon de mai jos

```
sablon struc
    _bp    dw ?
    _cs_ip dd ?
    baza  dw ?
    n      dw ?
sablon ends
```

Implementarea este următoarea (se consideră că stiva este descărcată de programul apelant):

```
put_base proc far
    push  bp          ; Secvența standard
    mov   bp, sp
    cmp   word ptr [bp].baza, 2  ; Test baza corectă
    jb    err_exit ; baza < 2
    cmp   word ptr [bp].baza, 36
```

Gheorghe Muscă – Programare în Limbaj de Asamblare

```

        ja      err_exit           ; baza > 36
;
push  ax                  ; Salvări
push  dx                  ; registre
mov   ax, [bp].n          ; Test
cmp   ax, [bp].baza       ; n < baza ?
mov   dl, al              ; x <-- n
jb    p_b_1               ; Salt la afişare
xor   dx, dx              ; Calcul n/baza
div   [bp].baza           ; şi n MOD baza
                                ; AX = n/baza
                                ; DX (DL) = n MOD baza
;
; Pregătire parametri pentru apelul recursiv
;
push  ax                  ; n/baza
push  [bp].baza           ; baza
call  far ptr put_base   ; Apel recursiv
add   sp, 4               ; Descărcare stivă
p_b_1:
    mov   ah, 2              ; Cod funcție DOS
    cmp   dl, 10             ; x < 10 ?
    jae  p_b_2               ; x + '0'
    add   dl, '0'            ; x + 'A' - 10
p_b_2:
    add   dl, 'A' - 10
p_b_3:
    int   21H                ; Afisare
    pop   dx                ; Refaceri
    pop   ax                ; registre
err_exit:
    pop   bp
    ret
put_base endp

```

Deoarece toate variabilele implicate sunt fără semn, se folosesc salturi condiționate de tip "below" și "above".

Un mic program de test (listat în continuare) citește un număr și o bază de la consolă și afișează numărul în baza dată, după care reia procesul de citire. Dacă numărul introdus este 0, programul se termină.

```

.model large
include io.h
.stack 1024
.code
start:
    init_ds_es
reluare:
    putsi  <cr, lf, 'Numar = '> ; Prompt
    getu
    or     ax,ax              ; Test oprire ?
    jz    exit                ; Salt la ieșire
    push  ax                  ; Plasare n în stivă
    putsi  <'Baza = '>        ; Prompt
    getu
    push  ax                  ; Plasare baza în stivă
    call  far ptr put_base   ; Apel
    add   sp, 4               ; Descărcare stivă

```

```

        jmp    reluare          ; Reluare
exit:
        exit_dos
end start

```

Un caz deosebit îl constituie funcțiile recursive, adică procedurile care întorc valori. Trebuie acordată atenție memorării valorilor întoarse de apelurile recursive. Să considerăm o funcție recursivă care întoarce numărul Fibonacci de ordinul n, definită recusiv prin:

```

fib (0) = fib (1) = 1
fib (n) = fib (n-1) + fib (n-2), pentru n >= 2

```

Această funcție a fost evaluată prin iterație în §2.4.4. Vom studia acum implementarea recursivă.

Se observă că, pentru valoarea lui fib(n), sunt necesare două apeluri recursive și salvarea rezultatului întors de primul apel. Presupunem că parametrul n se transmite prin stivă, că procedura este de tip FAR, că întoarce rezultatul în AX (ca număr fără semn) și că stiva este descărcată de programul apelant. Şablonul de acces este definit prin structura:

```

sablon struc
    _bp    dw ?
    _cs_ip dd ?
    n      dw ?
sablon ends

```

iar implementarea este următoarea:

```

fib proc far
;
; Primeste în stivă numărul n
; Întoarce în AX valoarea fib (n)
;
push  bp
mov   bp, sp
push  bx          ; Salvare registru folosit
;
mov   ax, [bp].n      ; Preia argumentul n
cmp   ax, 1          ; Cazul de bază: n <= 1
jbe   fib_1          ; Salt la evaluare imediată
;
dec   ax            ; Pregătire apel
push  ax            ; pentru n-1
call  fib            ; AX = fib (n-1)
add   sp, 2          ; Descărcare stivă
mov   bx, ax          ; Rezultat intermedian
;
mov   ax, [bp].n      ; Pregătire
dec   ax            ; apel
dec   ax            ; pentru
push  ax            ; n-2
call  fib            ; AX = fib (n-2)
add   sp, 2          ; Descărcare stivă
;
; În acest moment, AX = fib (n-2), BX = fib (n-1)
;
add   ax, bx          ; Calcul fib (n)
jmp   fib_2
fib_1:

```

```

        mov  ax, 1          ; Evaluare fib (0) și fib (1)
fib_2:
        pop  bx          ; Restaurare
        pop  bp          ; Revenire
fib endp
    
```

În implementarea de mai sus, este esențială salvarea și restaurarea registrului BX, care memorează rezultatul intermediar fib (n-1). Programul de test al acestei proceduri este lăsat ca exercițiu pentru cititor.

Nu putem încheia discuția despre proceduri recursive fără a vorbi despre problema clasică a turnurilor din Hanoi. În această problemă, se consideră n discuri de diametre distincte, așezate pe un suport vertical (turn sursă), în ordinea descrescătoare a diametrelor (vezi Figura 6.8). Se cere să se mute întreg ansamblul pe un alt turn (destinație), folosind un turn de manevră. Discurile se pot muta unul câte unul, cu restricția că un disc nu se poate așeza decât deasupra unui disc de diametru mai mare.




Figura 6.8 Turnurile din Hanoi

Algoritmul care rezolvă această problemă este în mod natural recursiv și se bazează pe observația că, dacă $n = 1$, adică există un singur disc, acesta se poate muta direct pe turnul destinație. Dacă notăm simbolic operația de mutare prin:

```
move (n, sursă, destinație, manevră)
```

unde sursă, destinație și manevră sunt cele trei turnuri, algoritmul se implementează prin:

```

move (n, sursă, destinație, manevră)
{
    dacă (n = 1)
        Mută discul de pe sursă pe destinație
    altfel {
        move (n-1, sursă, manevră, destinație)
        move (1, sursă, destinație, manevră)
        move (n-1, manevră, destinație, sursă)
    }
}
    
```

Se mută primele $n-1$ discuri de pe turnul sursă pe turnul manevră, folosind ca spațiu de lucru turnul destinație. În acest moment, turnul sursă conține un singur disc, turnul destinație este liber iar turnul manevră conține $n-1$ discuri.

Se mută unicul disc de pe turnul sursă pe turnul destinație. În acest moment turnul sursă este liber, turnul destinație conține discul cu diametrul cel mai mare iar turnul manevră conține celelalte $n-1$ discuri. Se aplică recursiv algoritmul, mutând $n-1$ discuri de pe turnul manevră pe turnul destinație și folosind turnul sursă ca spațiu de lucru.

Pentru un n dat, se deplasează explicit discul cu diametrul cel mai mare, ceea ce asigură îndeplinirea restricției din enunț.

Implementarea algoritmului se va face prin mesaje explicite la consolă, turnurile fiind codificate prin caractere.

Implementarea algoritmului, împreună cu un program de test, este listată în continuare. Structura şablon defineşte ordinea de plasare a parametrilor în stivă: n , sursă, destinație, manevră, toate de tip word. Toate apelurile recursive vor utiliza această ordine.

Simularea mutărilor se face prin afișarea unui mesaj, definit în zona de date. Câmpurile sursa_asc și dest_asc sunt variabile și vor identifica (prin câte un caracter) turnurile respective. Când $n = 1$ (cazul de bază), se depun caracterele care identifică turnurile în sirul de caractere mesaj și se afișează acest sir. Implementarea apelurilor recursive constă în plasarea în stivă a parametrilor în ordinea precizată și apelul procedurii.

Programul principal citește ciclic de la consolă numărul n , afișează un mesaj de identificare, pregătește parametrii în stivă și apelează procedura move. Bucla se oprește când se introduce valoarea 0 pentru n .

```
.model large
include io.h
.stack 4096
sablon struc
    _bp      dw ?
    _cs_ip   dw 2 dup (?)
    manevra  dw ?
    dest     dw ?
    sursa    dw ?
    n        dw ?
sablon ends
.data
    mesaj      db cr, lf, 'Muta discul de pe turnul '
    sursa_asc  db ?
                db ' pe turnul '
    dest_asc   db ?
                db 0
.code
move proc far
;
; Muta n discuri de pe sursa pe dest
; folosind manevra ca zonă de lucru
;
    push  bp
    mov   bp, sp
    cmp   word ptr [bp].n, 1           ; Test caz de bază (n = 1)
    jne   move_1                      ; Salt la cazul general
;
; Mutare explicită
;
```

Gheorghe Muscă – Programare în Limbaj de Asamblare

```
        mov  bx, [bp].sursa          ; Pregătire mesaj
        mov  sursa_asc, bl          ; Disc sursă
        mov  bx,[bp].dest           ; Disc destinație
        mov  dest_asc,bl            ; Afișare mesaj
        puts mesaj
        pop  bp
        retf 8                     ; Revenire cu descărcare
move_1:
;
; Cazul general
;
        mov  ax, [bp].n             ; Pregătire parametri
        dec  ax                     ; n-1
        push ax
        push [bp].sursa            ; Sursă
        push [bp].manevra          ; Actuala manevră devine
                                ; destinație
        push [bp].dest              ; Actuala destinație
                                ; devine manevră
        call far ptr move         ; Primul apel recursiv
;
        mov  ax, 1                  ; Pregătire parametri
        push ax
        push [bp].sursa            ; Sursă
        push [bp].dest              ; Destinație
        push [bp].manevra          ; Manevră
        call far ptr move         ; Al doilea apel recursiv
;
        mov  ax, [bp].n             ; Pregătire parametri
        dec  ax
        push ax ; n-1
        push [bp].manevra          ; Actuala manevră devine
                                ; sursă
        push [bp].dest              ; Destinație
        push [bp].sursa            ; Actuala sursă devine
                                ; manevră
        call far ptr move         ; Al treilea apel recursiv
;
        pop  bp
        retf 8                     ; Revenire cu
                                ; descărcare
move endp
;
; Program principal
;
start:
    init_ds_es
reluare:
    putsi <cr,lf,'Numar discuri: '>
    geti
    or   ax, ax
    jz   exit                   ; n = 0, stop
    putsi <cr,lf,'Muta '>
    puti ax ; Afișare n
    putsi <' turnuri, de pe A pe B,'>
    putsi <' folosind C ca'>
    putsi <' manevra', cr, lf>
;
    push ax                     ; n
    mov  ax, 'A'
```

```
    push  ax          ; Sursă
    mov   ax, 'B'
    push  ax          ; Destinație
    mov   ax, 'C'
    push  ax          ; Manevră
    call  far ptr move ; Apel
    jmp   reluare    ; Reluare
exit:
    exit_dos
end start
```

La implementarea algoritmilor recursivi, trebuie făcută o estimare a spațiului necesar de stivă. Spre exemplu, procedura move de mai sus necesită n nivele de apel recursiv. La fiecare nivel, se consumă 12 octeți din stivă (vezi structura şablon). Ca atare, volumul necesar de stivă este de ordinul $12 \times n$. Dacă estimăm că, într-o aplicație dată, valoarea lui n nu va crește peste 100, stiva acelei aplicații trebuie să fie de cel puțin 1200 de octeți.

Înainte de a trece la implementarea unui algoritm recursiv, trebuie să ne convingem (uneori prin demonstrații teoretice complicate), că algoritmul se termină (se atinge cazul de bază) după un număr finit de pași. Nu toate funcțiile definite recursiv au această proprietate. De exemplu, funcția f, "definită" pentru orice n pozitiv prin:

$$\begin{aligned}f(n) &= 5, \text{ pentru } n = 7 \\f(n) &= f(f(n+2)), \text{ pentru } n \text{ diferit de 7}\end{aligned}$$

nu este nici măcar o funcție corect definită. Încercarea de a evalua $f(5)$ conduce imediat la o tautologie ($f(5)$ depinde de $f(5)$), iar pentru n diferit de 5 și de 7 se obține un ciclu infinit.

Complexitatea algoritmilor recursivi este un alt subiect important, dar care depășește cadrul lucrării de față.

Trebuie deci acordată atenție descrierii teoretice a algoritmului recursiv, înainte de a trece la implementare. Observațiile despre implementarea algoritmilor recursivi sunt valabile și în cazul limbajelor de nivel înalt.

6.8 Proceduri cu număr variabil de parametri

Dezvoltarea procedurilor cu număr variabil de parametri are ca scop, pe de o parte, punerea în evidență a flexibilității limbajului de asamblare și, pe de altă parte, ilustrarea modului de implementare a unui mecanism care există în anumite limbaje de nivel înalt (de exemplu în C).

Dezvoltarea unei proceduri cu număr variabil de parametri se bazează pe următoarele reguli:

- parametrii se transmit prin stivă;
- există un număr (cel puțin 1) de parametri fischi (care nu pot lipsi) și care se transmit în vârful stivei (imediat după adresa de revenire);
- unul din parametrii care sunt totdeauna prezenti conține informații (codificate într-o formă oarecare) despre numărul și tipul parametrilor actuali;
- în procedură se analizează parametrii fischi, deducându-se numărul și tipul parametrilor variabili, care sunt apoi extrași din stivă;
- parametrii variabile ca număr transmiși în stivă trebuie să corespundă cu descrierea lor în parametrii fischi;
- stiva este descărcată de către programul apelant.

Regulile de mai sus nu sunt alese întâmplător. Astfel, faptul că parametrii fișă se găsesc în vârful stivei face ca aceștia să fie accesibili prin metoda standard. Programul apelant este cel care știe precis câți octeți a plasat în stivă înainte de apel, deci este natural ca acesta să descarce stiva.

Se înțelege acum de ce, în limbajul C, ordinea plasării argumentelor unei proceduri în stivă este de la dreapta la stânga: în acest fel, la o procedură cu număr variabil de parametri, parametrii fișă (care se află la începutul listei de argumente) vor fi plasați imediat după adresa de revenire.

Imaginea stivei la intrarea în procedură este ilustrată în Figura 6.9.




Figura 6.9 Stiva la intrarea într-o procedură cu număr variabil de parametri

Să considerăm un exemplu tipic de procedură cu număr variabil de parametri: o procedură printf_proc de afișare cu format. Există un parametru fix, de tip adresă NEAR a unui sir de caractere care descrie formatul de afișare. Sirul de caractere poate conține:

- caractere obișnuite, care se afișează ca atare;
- descriptori de format, compuși din caracterul % și un al doilea caracter care descrie tipul afișării:
 - ◆ %s pentru siruri de caractere;
 - ◆ %d pentru întregi pe 16 biți cu semn;
 - ◆ %u pentru întregi pe 16 biți fără semn.
- secvențe escape:
 - ◆ '\n' pentru CR și LF;
 - ◆ '\\' pentru caracterul '\\';
 - ◆ '%%' pentru caracterul '%';

Pentru fiecare specificator de format, se presupune că există un parametru variabil în stivă, care trebuie afișat corespunzător. Pentru întregi, se transmite în stivă conținutul care trebuie afișat, iar pentru siruri, se transmite adresa de început, de tip NEAR. Se observă că, în toate situațiile, se trasmite în stivă un cuvânt (word), ceea ce facilitează implementarea. Dacă dimensiunile parametrilor variabili ar fi diferite, ar trebui folosit operatorul de conversie de tip PTR, pentru a încărca parametrul în mod corect.

Pentru afişarea propriu zisă se vor folosi macroinstructiunile de apel puts (şiruri), puti (întregi cu semn), putu (întregi fără semn) și putc (afişare caracter), descrise în Capitolul 2.

Procedura și toate adresele implicate se consideră de tip NEAR. În Figura 6.10 este descris conținutul stivei la un apel tipic al acestei proceduri.




Figura 6.10 Conținutul stivei la intrarea în printf_proc

Pentru accesul la stivă se va utiliza şablonul:

```
sablon struc
    _bp      dw ?
    _ip      dw ?
    format   dw ?          ; Adresă sir format
    par      dw ?          ; Primul parametru
sablon ends
```

Procedura va consta dintr-o buclă de analiză a caracterelor din sirul format, a cărui adresa e gestionată în BX. Astfel, [BX] va reprezenta caracterul curent. Pentru specificatorii de format, se gestionează un contor de parametri variabili, în registrul DI, care va lua valorile 0, 2, 4 etc. (toți parametrii variabili sunt pe 16 biți). Accesul la parametrii variabili se va face prin expresia [BP][DI].PAR, deci printr-o adresare bazată și indexată, cu deplasament. Se tratează corespunzător secvențele escape (care încep cu \) și specificatorii de format (care încep cu %), ambele situații presupunând citirea unui caracter suplimentar din sirul care descrie formatul. Dacă parametrii nu ar avea toți aceeași dimensiune, s-ar utiliza expresii de genul DWORD PTR [BP][DI].PAR, caz în care registrul DI ar fi actualizat cu dimensiunea parametrului (în cazul de față, cu 4).

Implementarea este următoarea:

```
printf_proc proc near
    push bp
    mov bp,sp
    mov bx,[bp].format
    mov di,0
    ; Salvare BP
    ; Adresă sir format
    ; Contor parametri
    ; variabili

pr_loop:
    mov al,[bx]
    test al,al
    ; Caracter curent
    ; Test sfârșit de sir
```

```

        jnz    aici1                                ; care descrie formatul
        jmp    pr_end

aici1:
        cmp    al, '%'                            ; Test spec. format
        jne    pr_char                            ; Nu este spec.
        inc    bx                                 ; Dacă da, se citește
        mov    al, [bx]                            ; caracterul următor
        cmp    al, 's'                            ; Este %s ?
        je     pr_str                             ; Da, salt la sfîșare
        cmp    al, 'd'                            ; Este %d ?
        je     pr_int                             ; Da, salt la sfîșare
        cmp    al, 'u'                            ; Este %u ?
        je     pr_u                               ; Da, salt la sfîșare
        cmp    al, '%'                            ; Este %% ?
        je     aici2                             ;
        jmp    pr_err                            ; Nu a fost nici unul
                                                ; din specificatorii
                                                ; așteptați: se afișează
                                                ; un mesaj de eroare

aici2:
        putc   al                                 ; Afîșare caracter
        inc    bx                                 ; Avans adresă sir
        jmp    pr_loop                           ; Reluare

pr_char:
        cmp    al, '\'                            ; Este secvență escape ?
        je     pr_esc                            ; Da, salt la tratare
        putc   al                                 ; Dacă nu, e vorba de
                                                ; un car. obișnuit, care
        inc    bx                                 ; se afișează și se reia
        jmp    pr_loop                           ; bucla de parcurgere

pr_esc:
        inc    bx                                 ; Tratare secvență escape
        mov    al, [bx]                            ; Luăm următorul caracter
        cmp    al, '\n'                            ; Este \n ?
        jne    esc1                             ; Nu, salt
        putc   cr                               ; Da, se afișează
        putc   lf                               ; CR LF și se reia
        inc    bx                                 ; bucla de
        jmp    pr_loop                           ; parcurgere

esc1:
        cmp    al, '\\'                            ; Este \\
        jne    pr_err                            ; Nu, secvență escape
                                                ; greșită
        putc   al                                 ; Da, afișează \
        inc    bx                                 ; și reia bucla
        jmp    pr_loop                           ; de prelucrare

;
; Afîșarea parametrilor variabili
;

pr_str:
        push   bx                                ; Afîșare sir. Preluăm
        mov    bx, [bp+di].par                   ; adresa din stivă și
        puts   [bx]                             ; și afișăm sirul (cu
        pop    bx                                ; salvare/restaurare BX

pr_next:
        inc    bx                                ; Car. următor în format
        add    di, 2                             ; Contor par. variabili
        jmp    pr_loop                           ; Reluare

```

```

pr_int:
    puti  [bp+di].par          ; Întreg cu semn
    jmp   pr_next              ; Reluare
pr_u:
    putu  [bp+di].par          ; Întreg fără semn
    jmp   pr_next              ; Reluare
pr_err:
    putsi <cr, lf, 'printf: Eroare format', cr, lf> ; Mesaj de eroare
pr_end:
    pop   bp                  ; Refacere BP
    ret               ; Revenire
printf_proc endp

```

Ne propunem acum să dezvoltăm o macroinstructiune de apel, numită printf, care să se apropie cât mai mult de un limbaj de nivel înalt, deci să putem scrie în textul sursă ceva de genul:

```
printf <"Numerele m n și p sunt: %d %d %d\n">, <m, n, p>
```

Macroinstructiunea va avea deci un prim parametru (șirul care descrie formatul) și o listă de parametri variabili (care poate chiar lipsi):

```
printf macro format, param
```

Scrierea unei astfel de macroinstructiuni ridică probleme deosebite, dar pe care operatorii și directivele limbajului de asamblare le pot rezolva.

Șirul care descrie formatul se definește în segmentul de date, printr-o tehnică similară celei de la macroinstructiunea putsi, descrisă în Capitolul 5.

Problema cea mai dificilă este că parametrii variabili trebuie puși în stivă în ordine inversă decât apar în lista din apelul macroinstructiunii. Lista poate fi parcursă cu instrucțiunea IRP, dar numai de la stânga la dreapta.

Pentru a pune parametrii în ordinea dorită, se va parcurge lista lor de două ori. La prima parcurgere, se contorizează numărul de parametri (cu operatorul de atribuire =) și se creează spațiu în stivă, prin secvența:

```

N = 0
irp   x, <param>
      sub sp, 2
      N = N + 1
endm

```

La a doua parcurgere, se plasează parametrii în stivă. Problema este că indicatorul SP trebuie să fie incrementat cu 2, deci nu se poate utiliza o simplă instrucțiune PUSH. Se execută deci secvența:

```

irp   x, <param>
      add  sp, 2
      push x
      add  sp, 2
endm

```

care pune parametrii x în stivă, de sus în jos (vezi Figura 6.11).




Figura 6.11 Plasarea parametrilor în stivă la macroinstructiunea printf

Toată această secvență se execută numai dacă lista param este nevidă. După depunerea parametrilor variabili, se decrementează SP cu $2 \cdot N$ (deci se aduce SP pe primul parametru), se depunde în stivă adresa formatului și se apelează procedura printf_proc. Contorizarea parametrilor în constanta simbolică N asigură și o descărcare comodă a stivei, prin adunarea valorii $2 \cdot (N+1)$. Dacă sirul format lipsește, se forțează o eroare de asamblare.

Implementarea completă a macroinstructiunii este următoarea:

```

printf macro format, param
local loc_for
ifb    <format>
      %out : Lipsa format....
      .err
      exitm
endif
.data
loc_for    db    format      ;; Definire sir format
           db    0
.code
N = 0          ;; Contor par. variabili
ifnb   <param>        ;; Dacă există par. var.
      irp   x, <param>
            sub   sp, 2      ;; Creare spațiu
            N = N + 1        ;; Contor
endm

      irp   x, <param>
            add   sp, 2      ;; În jos
            push  x          ;; În sus
            add   sp,2        ;; În jos
endm
      sub   sp, 2*N        ;; SP pe primul par. var.
endif
lea   ax, loc_for       ;; Adresă format
push  ax               ;; Depunere în stivă
call  printf_proc       ;; Apel
add   sp, 2*(N+1)       ;; Descărcare stivă
endm

```

Să considerăm un exemplu de apel, în care avem definite datele:

```

.data
n1    dw    1234
n2    dw    -3
n3    dw    4321

```

```
sir1    db      "abcdef",0
sir2    db      "1234567890",0
sir3    db      "Acesta e un sir...",0
s1      dw      sir1
s2      dw      sir2
s3      dw      sir3
```

Secvența de test (programul principal) este următoarea:

```
.code
start:
    init_ds_es
    printf <"Numarul 1: %d\n">, <n1>
    printf <"Numerele 2 si 3: %d %d">, <n2,n3>
    printf <"\nSirul 1: %s %% \\ %%">, <s1>
    printf <"\nSirul 2: %s\nSirul 3: %s">, <s2,s3>
    printf <"\nDoar format (fara descriptori) \n">
    printf <"n2 cu semn = %d, n2 fara semn = %u\n">, <n2,n2>
    exit_dos
end start
```

Se observă necesitatea definirii unor variabile pointer (s2, s2, s3) care conțin adresele sirurilor de caractere sir1, sir2 și sir3. Rularea acestui exemplu produce ieșirea la consolă:

```
Numarul 1: 1234
Numerele 2 si 3: -3 4321
Sirul 1: abcdef % \ %
Sirul 2: 1234567890
Sirul 3: Acesta e un sir...
Doar format (fara descriptori)
n2 cu semn = -3, n2 fara semn = 65533
```

Mai trebuie observat că tehnica de transfer a parametrilor variabili, expusă mai sus, asigură și o bună protecție a controlului programului în situații de eroare, adică atunci când descriptorii de format nu coincid cu parametrii, atât ca număr cât și ca tip. Să considerăm, de exemplu, apelurile:

```
printf <"Ceva gresit %d %d %d\n">, <n1, n2>
printf <"Ceva si mai gresit %d %d\n">, <n1, s2, s3>
```

În primul caz, formatul precizează că în stivă sunt trei parametri de tip întreg, dar la apel nu se transmit decât doi. Procedura va afișă un al treilea parametru fictiv dar, deoarece stiva este descărcată de programul apelant, revenirea se face corect și programul nu se distrugе. În al doilea caz, primul parametru se afișează corect, al doilea (o adresă de sir) este afișat incorrect ca întreg, iar al treilea este pur și simplu ignorat. Si în acest caz, revenirea în programul apelant și continuarea să nu sunt compromise.

6.9 Tehnici avansate de programare

În subcapitolele anterioare au fost descrise elementele de bază ale programării în ASM (acțiunile principale din programarea structurată, proceduri, definiții de date etc.). Cu aceste elemente de bază, se pot aplica orice tehnici de programare cunoscute de la limbaje de nivel înalt. Vom exemplifica o asemenea tehnică, anume cea a automatelor de stare.

Considerăm problema elementară a numărării cuvintelor dintr-un sir de caractere. Se consideră o serie de caractere speciale, numite delimitatori (în

cazul de față: spațiu, punct, virgulă, semn de întrebare, semn de exclamare și cratimă. Evident, se poate specifica orice fel de delimitatori.

Incrementarea numărului de cuvinte nu depinde numai de caracterul curent din sir. De exemplu, un caracter care nu este delimitator va conduce la incrementarea numărului de cuvinte numai în cazul în care caracterul anterior a fost un delimitator. Acest tip de decizie, care se bazează pe o informație anterioară, sugerează utilizarea unei variabile de stare, care să memoreze starea curentă a prelucrării.

În cazul de față, vom avea două stări: în interiorul unui cuvânt și în exteriorul unui cuvânt. Notăm cele două stări cu IN_W și OUT_W. Pentru memorarea lor este suficientă o variabilă de stare binară.

Algoritmul se modelează ca un automat, care acceptă intrări și generează ieșiri. Funcție de intrări și de starea curentă, se trece într-o stare următoare, emițând sau nu o ieșire. În cazul de față intrările sunt:

- in_1 = caracterul curent din sir nu este delimitator;
- in_2 = caracterul curent din sir este delimitator.

Ieșirea automatului este:

- out = incrementeză numărul de cuvinte

Trebuie acum să codificăm stările, intrările și ieșirile. Pentru simplitate, vom utiliza registrele procesorului:

- stare = IN_W: CX = 1
- stare = OUT_W: CX = 0
- in_1: BX = 1
- in_2: BX = 0
- out: Incrementeză AX.

Descriem acum funcționarea automatului, prin diagrama din Figura 6.12.




Figura 6.12 Descrierea funcționării unui automat de stare

Dacă suntem în interiorul unui cuvânt (starea IN_W), atunci un delimitator (intrarea in_2) provoacă trecerea în exteriorul unui cuvânt (starea OUT_W) iar un caracter care nu este delimitator (intrarea in_1) menține starea IN_W. Dacă suntem în exteriorul unui cuvânt (starea OUT_W), atunci un delimitator (intrarea in_2) menține starea OUT_W iar un caracter care nu este delimitator (intrarea in_1) provoacă trecerea în starea IN_W și forțarea semnalului de

ieșire out. Inițial, automatul se află în starea OUT_W. Se remarcă faptul că, în codificarea aleasă, starea următoare este dictată de codificarea intrării (mai precis, comutarea stării va însemna o simplă instrucțiune MOV CX, BX).

Evoluția variabilei de stare, a intrărilor in_1 și in_2 precum și a ieșirii out în cazul prelucrării sirului de caractere "Un text oarecare." sunt ilustrate în Figura 6.13.




Figura 6.13 Evoluția automatului de stare în cazul sirului "Un text oarecare."

Implementarea algoritmului urmărește îndeaproape funcționarea automatului și constă în principal din:

- evaluarea intrărilor;
- comutarea stării;
- forțarea ieșirii.

Presupunem că adresa sirului este transmisă procedurii în DS:SI și că rezultatul este întors în AX. Implementarea este următoarea:

```

count proc near
    push cx           ; Salvări
    push bx           ; registre
    push si
    mov  cx, 0         ; Stare inițială = OUT_W
    mov  ax, cx         ; Număr de cuvinte = 0
    reluare:
        cmp byte ptr [si], 0      ; Test sfârșit de sir
        je  gata
        cmp byte ptr [si], ' '
        je  et_1
        cmp byte ptr [si], tab
        je  et_1
        cmp byte ptr [si], '.'
        je  et_1
        cmp byte ptr [si], ','
        je  et_1
        cmp byte ptr [si], '?'
        je  et_1
        cmp byte ptr [si], '!'
        je  et_1
        cmp byte ptr [si], '-'
        je  et_1
    gata:
        mov  ax, count
        pop  bx
        pop  cx
        ret
et_1:
    inc  si
    mov  cx, cx
    jmp  reluare

```

```

        mov  bx, 1                      ; Nu e delimitator
        jmp  et_2
et_1:   mov  bx, 0                      ; Delimitator
;
;     În acest moment, intrările sunt poziționate
;     BX = 1 (in_1) sau BX = 0 (in_2)
et_2:   cmp  cx, 0                      ; Stare = OUT_W ?
        je   et_3
        mov  cx, bx
        jmp  et_5
et_3:   cmp  bx, 1                      ; In starea OUT_W si
        jne  et_4
        inc  ax                         ; cu intrarea in_1,
                                    ; se genereaza iesirea
                                    ; out (incrementeaza nr.
                                    ; de cuvinte)
et_4:   mov  cx, bx
et_5:   inc  si                         ; Trecere la
        jmp  reluare                   ; caracterul următor
gata:  pop  si                         ; Refaceri
        pop  bx                         ; registre
        pop  cx
        ret                           ; Rezultat în AX
count endp

```

Un program de test al procedurii count este listat mai jos. Se citesc de la consolă siruri de caractere (până la introducerea sirului vid), se apelează procedura count și se afișează numărul de cuvinte din sirul introdus.

```

.model    small
include  io.h
.data
        sir    db     80 dup (0)          ; Spațiu de lucru
.stack 1024
.code

start:  init_ds_es
iar:    putsi  <'Introduceti un sir: '>
        gets   sir
        mov    al, sir
        test   al, al
        jz    exit
        putsi  <'Sirul are '>
        lea    si, sir
        call   count
        puti   ax
        putsi  <' cuvinte', cr, lf>
        jmp   iar
exit:   exit_dos
end    start

```

6.10 Programarea coprocesoarelor matematice

În Capitolul 2 a fost prezentat setul de instrucțiuni al coprocesoarelor matematice 80x87. Coprocesorul recunoaște variabile întregi și toate cele trei tipuri de numere reale (precizie simplă, dublă și extinsă).

În acest subcapitol ne propunem să dezvoltăm un set de proceduri și macroinstructiuni care să rezolve problema conversie din formatul real (în simplă precizie), în format ASCII (șir de caractere), ceea ce va permite apoi și introducerea și afișarea de numere reale la consolă.

6.10.1 Conversia ASCII-real

Dezvoltăm o procedură de tip far, cu numele ATOF_PROC, care primește în stivă o adresă near a unui sir de caractere și o adresă near a unui număr real. Şablonul de acces la stivă este:

```
sablon_atof struc
    dw      ?      ; Loc pentru BP
    dd      ?      ; Loc pentru adresa de revenire
    adr_num   dw    ?
    adr_sir    dw    ?
sablon_atof ends
```

Procedura de conversie se bazează pe următorul algoritm:

```
valoare = 0;
divizor = 1.0
punct = FALSE;
minus = FALSE;
preia caracter din sirul sursă;
if (carcater == '-') {
    minus = TRUE;
    preia următorul caracter din sirul sursă;
}
while (caracterul curent este cifră zecimală sau '.') {
    if (caracter == '.')
        punct = TRUE;
    else {
        cifra = caracter - '0';
        cifra_float = (float) cifra;
        valoare = 10.0 * valoare + digit_float;
        if (punct)
            divizor = divizor * 10.0;
    }
    preia următorul caracter din sir;
}
if (punct)
    valoare = valoare/divizor;
if (minus)
    valoare = - valoare;
return valoare;
```

Ideea generală este calculul numărului ca și când nu ar avea punctul zecimal și împărțirea apoi la o putere a lui 10, calculată după numărul de zecimale. De exemplu, sirul 123.456 va conduce la calculul valorii 123456, care se împarte apoi la 1000.

Implementarea este următoarea:

```

.data
    _zece      dd  10.0
    _punct     db  ?
    _minus     db  ?
    _cifra     dw  ?

.code
    atof_proc proc far
    push  bp
    mov   bp, sp
    push  bx
    push  si
    ; Salvări
    ; registre

    fld1
    fldz
    ; ST ← 1.0
    ; ST ← 0.0, ST(1) ← 1.0

;
; Asignăm valoare în ST și divizor în ST(1)
;

    mov  _punct, 0
    mov  _minus, 0
    xor  bh, bh
    ; Necesar la depunere întreg

;
    mov  si, [bp].adr_sir
    cmp  byte ptr [si], '-'
    jne  atof_1
    mov  _minus, 1
    inc  si

atof_1:
    mov  bl, [si]
    cmp  bl, '.'
    jne  atof_2
    mov  _punct, 1
    jmp  atof_3
    ; Salt la reluare

atof_2:
    cmp  bl, '0'
    jb   atof_4
    cmp  bl, '9'
    ja   atof_4
    sub  bl, '0'
    mov  _cifra, bx
    fmul _zece
    fiadd _cifra
    ; Conversie la întreg
    ; Memorăm ca întreg
    ; valoare ← 10*valoare
    ; valoare ← valoare +cifra

;
    cmp  _punct, 1
    jne  atof_3
    fxch
    fmul  _zece
    fxch
    ; Test punct
    ; Schimbăm ST cu ST(1)
    ; pentru că vrem să înmulțim
    ; divizorul cu zece, apoi
    ; schimbăm la loc

atof_3:
    inc  si
    jmp  atof_1
    ; Reluare
    ; buclă

atof_4:
    fdivr
    ; Împărțire ST la ST(1) cu
    ; descărcarea stivei; Rezultat
    ; în ST

    cmp  _minus, 1
    jne  atof_5
    fchs
    ; Test semn
    ; Schimbare semn

atof_5:

```

```

mov  bx, [bp].adr_num          ; Adresă număr real
fstp dword ptr [bx]           ; Depunere ST cu descărcarea
fwait                         ; stivei, apoi FWAIT

pop   si                      ; Refaceri
pop   bx                      ; registre
pop   bp
retf                           ; Revenire
atof_proc endp

```

Se observă că stiva coprocesorului este lăsată în starea în care era la intrarea în procedură, lucru care trebuie avut în vedere la toate programele care lucrează cu 80x87. Într-o implementare mai pretențioasă, s-ar salva toate cele 8 registre ale coprocesorului ca variabile de tip TBytes, într-o zonă proprie de memorie și s-ar refața revenirea în programul apelant.

De asemenea, bucla de citire caractere ar trebui să testeze dacă s-a întâlnit deja punctul zecimal. Pentru simplitate, s-a presupus aprioric că sirul sursă conține cel mult un punct zecimal.

Scriem acum o macroinstructiune de apel, cu numele atof, care are ca parametri sirul de caractere sursă și variabila reală în care se va depune rezultatul. Ambii parametri sunt folosiți în instrucțiuni LEA, care preiau adresele near și le transmit către procedura atof_proc.

```

atof  macro a_sir, a_num
      push  si
      lea   si, a_sir
      push  si
      lea   si, a_num
      push  si
      call  far ptr atof_proc
      add   sp, 4
      pop   si
endm

```

Dispunând de macroinstructiunea atof, putem scrie o macroinstructiune cu numele getf, care să citească un număr real de la consolă. Operația se compune dintr-o citire de sir de caractere și o conversie ASCII-real. Parametrul macroinstructiuni este variabila dword în care se depune numărul real citit.

```

getf  macro a_num
local sir
.data
      sir    db 80 dup (0)
.code
      gets  sir
      atof  sir, a_num
endm

```

6.10.2 Compararea numerelor reale

O problemă importantă care se pune este compararea a două numere reale. Procesorul dispune de instrucțiuni de comparare, în urma cărora se poziționează o serie de flaguri interne din cuvântul de stare (Status Word). Acest cuvânt poate fi depus în memorie cu instrucțiuni FSTSW și citit apoi prin instrucțiuni 80x86. Preferăm totuși o altă abordare, mai simplă, anume efectuarea diferenței celor doi operanți și poziționarea corespunzătoare a bistabililor ZF și CF din 80x86. Astfel, se vor putea utiliza instrucțiuni de slat

condiționat specific 80x86 (JE, JB, JGE etc.). Testarea semnului diferenței se face prin examinarea fizică a bitului de semn din reprezentarea internă a numerelor în simplă precizie. Testarea cazului de egalitate se face testând dacă toți cei patru octeți ai diferenței sunt nuli.

Dezvoltăm pentru început o primă variantă a unei macroinstrucțiuni cu numele F_COMP, având un singur operand, care compară numărul real din vârful stivei coprocesorului cu operandul specificat.

```
f_comp macro val
    local temp, et_gt, et_lt, et_eq, gata
.data
    temp dd ?
.code
    push ax ; Salvare AX
    fld st ; Se face o copie a vârfului stivei,
            ; pentru a nu altera ST inițial
    fsub dword ptr val ; ST ← ST - val
    fstp dword ptr temp ; Depunem diferența
    mov al, byte ptr temp+3 ; Luăm ultimul octet din reprezentare
    and al, 10000000B ; Filtru bit de semn
    jnz et_lt ; Diferență negativă ?
    mov ax, word ptr temp ; Nu, testăm dacă nu este zero
    or ax, word ptr temp+2 ; Zero real are toți cei 4 octeți nuli
    jz et_eq ; Este zero ?
et_gt: ; Nu, înseamnă că e mai mare
    mov ax, 2 ; Facem o comparație pentru a
    cmp ax, 1 ; poziționa indicatorii
    jmp gata
et_lt: ; Cazul mai mic
    mov ax, 1
    cmp ax, 2
    jmp gata
et_eq: ; Cazul egal
    mov ax, 1
    cmp ax, 1
    jmp gata
gata: ; Refacere AX și gata
    pop ax
endm
```

În această variantă, se efectuează o scădere temporară a celor doi operanzi și apoi se poziționează flagurile procesorului de bază, exact ca la o instrucțiune de comparație 80x86. Pentru claritate, s-au forțat comparațiile între valorile întregi 1 și 2.

O a doua variantă a acestei macroinstrucțiuni se bazează pe instrucțiunea FSTSW a coprocesoarelor incorporate în procesoarele 80486 și cele ulterioare. Această instrucțiune poate salva registrul de stare 80x87 (de 16 biți) în registrul AX al procesorului de bază. Este prima instrucțiune în care cele două procesoare comunică direct, și nu prin intermediul memoriei. În plus, cei 3 biți care raportează rezultatul comparației în registrul de stare 80x87 (din cei mai semnificativi 8 biți ai registrului de stare) corespund cu flagurile CF, ZF și PF din registrul de flaguri. Astfel, o instrucțiune SAHF va poziționa flagurile CF și ZF cu aceeași semnificație ca la comparațiile de întregi. Flagul PF setat va corespunde unor operanzi în virgulă mobilă necomparabili.

Concret, asta înseamnă că putem folosi salturi condiționate ce conțin "below" și "above" (JB, JBE, JA, JAE etc.)

De remarcat că noile procesoare Intel (Pentium Pro și ulterioare) dispun și de alte mecanisme de testare a rezultatului comparației a două valori reale. A se vedea secțiunile 7.3.3 și 7.5.6 din "Intel Architecture – Software Developers Manual", vol. I - "Basic Architecture".

A doua implementare a macroinstructiunii F_COMP este următoarea:

```
f_comp macro val
local temp
.data
    temp dw ?
        ; Spațiu de lucru
.code
    push ax
    fcom dword ptr val
    fstsw word ptr temp
;
;    fstsw ax
        ; Numai de la 80486 înapoi!
;
    fwait
    lahf
    mov ax, word ptr temp
    sahf
    pop ax
        ; Putem folosi JB, JBE, JA, JAE și JE
endm
```

6.10.3 Conversia real-ASCII

Dezvoltăm acum procedura ftoa_proc, care primește în stivă adresa near a unui număr real și adresa near a unui sir de caractere. Procedura convertește numărul real într-un sir de caractere care conține reprezentarea valorii reale în format științific:

```
x.xxxxxxE±yy
```

Accesul la parametrii din stivă se face prin structura şablon:

```
sablon_ftoa struc
    dw ?
    dd ?
    val dd ?
    siradr dw ?
sablon_ftoa end
```

Implementarea folosește o serie de constante și variabile de lucru, definite mai jos. Algoritmul de conversie este următorul:

```
;
;      val = Valoarea reală care se convertește
;      sir = Adresa sirului destinație
;
if (val < 0.0) {
    *s++ = '-';
    val = -val;
}
_exp = 0;
if (val != 0.0) {
    if (val >= 0.0) {
```

```

        do {
            val = val / 10.0;
            _exp = _exp + 1;
        } while (value >= 10.0);
    }
    else {
        while (val < 1.0) {
            val = val * 10.0;
            _exp = _exp - 1;
        }
    }
    val = val + 0.0000005;
    if (value > 10.0) {
        value = value / 10.0;
        _exp = _exp + 1;
    }
}
_cifra = (int) val;           /* Obligatoriu cu trunchiere, nu cu rotunjire */
*s++ = _cifra + '0';
*s++ = '.';
for (i = 1; i <= 6; i++) {
    val = 10.0 * (val - (float) _cifra);
    _cifra = (int) val;           /* Obligatoriu cu trunchiere, nu cu rotunjire */
    *s++ = _cifra + '0';
}
*s++ = 'E';
if (_exp < 0) {
    *s++ = '+';
    _exp = - _exp;
}
else
    *s++ = '+';
/* scrie _exp ca două cifre în sir */
*s = 0;                      /* Terminator de sir */

```

Se aduce numărul în intervalul [1.0, 10.0), prin împărțiri sau înmulțiri succesive cu 10. Înmulțirile sau împărțirile cu 10 sunt contorizate în variabila întregă `_exp` (exponent zecimal). În acest moment, partea întreagă a numărului are o singură cifră, care se poate extrage și depune în sir. Se determină partea fractionară, prin scăderea părții întregi, se înmulțește cu 10 și se reia generarea cifrelor. Acest proces se repetă de 5 ori, pentru cifrele de după virgulă. În prealabil, se realizează o rotunjire a numărului, prin adunarea valorii reale 0.000005.

Algoritmul depinde esențial de modul în care se face conversia real-întreg. Este obligatoriu ca această conversie să se realizeze prin trunchiere, nu prin rotunjire. De exemplu, valoarea 2.999 trebuie să genereze prima cifră 2 și nu 3.

Coprocesorul poate opera în mai multe moduri de rotunjire/trunchiere. Aceste moduri sunt controlate de biți 10 și 11 (câmpul RC sau Round Control), din cuvântul de control (Control Word). Pentru situația de față, este adecvat modul `RC = 1`. Procedura trebuie deci să poționeze câmpul RC din cuvântul de control, printre-o salvare/încărcare în memorie. La revenirea în programul apelant, se va refașă câmpul RC original. A se vedea secțiunea 7.3.4.3 din "Intel Architecture – Software Developers Manual", vol. I - "Basic Architecture".

Procedura necesită o serie de constante reale și variabile de manevră, definite mai jos. Numele variabilelor coincid cu cele de la descrierea algoritmului. Evident, vom aloca variabila reală în vârful stivei 80x87. Exponentul este gestionat ca număr întreg.

```
.data
    _zece dd    10.0
    _unu  dd    1.0
    _zero dd    0.0
    _rounddd dd  0.0000005
    _exp   dw    ?
    _cifra dw    ?
    _temp  dd    ?
    _cw    dw    ?
```

Implementarea propriu-zisă este următoarea:

```
.code
ftoa_proc proc far
    push    bp
    mov     bp, sp

    push    ax          ; Salvare
    push    bx          ; registre
    push    si
    push    cx

    fstcw   _cw          ; Salvare Control Word
    mov     ax, _cw
    and    ax, NOT 0000110000000000B ; Filtru biții 10 și 11
    or     ax, 0000010000000000B ; Forțare RC = 1
    mov     _cw, ax        ; Înapoi în memorie
    fldcw   _cw          ; Înapoi în 80x87

    mov     _exp, 0        ; Exponent
    mov     si, [bp].siradr ; Adresă sir de caractere
    fld    dword ptr [bp].val ; Valoare reală în ST
    f_comp _zero
    ja     ftoa_poz      ; Este > 0 ?
    mov     byte ptr [si], '-' ; Nu, depunem '-'
    inc     si             ; În sir și schimbăm
    fchs

ftoa_poz:
    f_comp _zero
    jne    ftoa_aici    ; Este diferit de 0.0 ?
    jmp    ftoa_1

ftoa_aici:
    f_comp _zece
    jnge   ftoa_2        ; Da, îl convertim
    ; Da, îl convertim

ftoa_3:
    fdiv   _zece
    add    _exp, 1        ; Nu, îl împărțim la 10
    f_comp _zece
    jnl    ftoa_3        ; și ținem minte la exponent
    ; până când ajunge
    ; mai mic strict decât 10

ftoa_2:
    f_comp _unu
    jge    ftoa_1        ; Este mai mare sau egal ca 1 ?
    fmul   _zece
    sub    _exp, 1        ; Da, salt
    ; Nu, îl înmulțim cu 10 și ținem minte
    ; la exponent, până ajunge
    jmp    ftoa_2        ; mai mare sau egal ca 1
```

```

ftoa_1:
    fadd      _round           ; Rotunjim la 7 zecimale
    f_comp    _zece            ; Poate a depășit acum 10 ?
    jna      ftoa_4            ; Dacă da, corectăm
    fdiv      _zece
    add      _exp, 1

ftoa_4:
    fist      _cifra           ; Generăm partea întreagă
    mov      bl, byte ptr _cifra
    add      bl, '0'
    mov      [si], bl          ; o convertim la caracter ASCII
    inc      si                ; și o depunem în sir
    mov      byte ptr [si], '.'
    inc      si                ; Apoi punctul zecimal

    mov      cx, 6             ; Buclă de 6 cifre după punct

ftoa_5:
    fisub    _cifra           ; ST = partea fractionară
    fmul    _zece            ; ST = ST * 10
    fist      _cifra           ; Partea întreagă
    mov      bl, byte ptr _cifra
    add      bl, '0'
    mov      [si], bl          ; Determinare cifră
    inc      si
    loop     ftoa_5

    fstp      _temp            ; Descarcăm stiva 8087
    mov      byte ptr [si], 'E' ; Notație științifică
    inc      si
    cmp      _exp, 0            ; Test semn exponent
    jge      ftoa_6
    mov      byte ptr [si], '-'
    neg      _exp              ; Negativ
    jmp      ftoa_7

ftoa_6:
    mov      byte ptr [si], '+'
    ; Pozitiv

ftoa_7:
    inc      si                ; Exponent
    mov      ax, _exp           ; AH = _exp / 10;
    aam
    or       ax, '00'           ; AL = _exp MOD 10
    mov      [si], ah
    inc      si                ; Ambele cifre convertite la ASCII
    mov      [si], al
    inc      si                ; Depunere în sir
    mov      byte ptr [si], 0   ; În fine, terminatorul de sir

    pop      cx                ; Refacere
    pop      si                ; registre
    pop      bx
    pop      ax

    pop      bp                ; Revenire
    retf

ftoa_proc endp

```

Se observă utilitatea foarte mare a macroinstructiunii f_comp, care face ca implementarea algoritmului să fie o simplă transpunere a descrierii de tip pseudo-cod.

Dezvoltăm acum o macroinstructiune de apel cu numele ftoa și cu parametri adecvați (valoarea reală și sărul destinație):

```
ftoa    macro val, sir
        push   ax
        lea    ax, sir
        push   ax
        mov    ax, word ptr val + 2
        push   ax
        mov    ax, word ptr val
        push   ax
        call   far ptr ftoa_proc
        add    sp, 6
        pop    ax
endm
```

Dispunând de macroinstructiunea ftoa, putem dezvolta o macroinstructiune putf, care să afișeze la consolă un număr real, prin conversie real-ASCII și afișare săr:

```
putf    macro val
        local  sir
        .data
                sir    db 30 dup (0)
        .code
                ftoa  val, sir
                puts  sir
endm
```

6.10.4 Un program de test

Dacă presupunem macroinstructiunile de mai sus definite într-un fișier float.h, putem scrie un mic program de test, care citește un număr real de la consolă și afișează valorile incremenate de 10 ori.

```
.model      large
.8087
include    io.h
include    float.h
.stack     1024
.data
        numar      dd    ?
        unu       dd    1.0
.code
start:
        init_ds_es
        finit
        putsi      <'Introduceti un numar real: '>
        getf
        numar
        putsi      <'Valorile incrementate sunt:', cr, lf>
        mov       cx, 10
iar:
        putf      numar
        putsi      <cr, lf>
        fld       numar
        fadd
        fstp      numar
```

```
loop      iar
        exit_dos
end      start
```

Se observă instrucțiunea FINIT, care initializează coprocesorul matematic. Directiva .8087 instruiește asamblorul să accepte mnemonice specifice coprocesorului 8087. Funcție de tipul coprocesorului și al procesorului de bază, se pot utiliza directivele .287, .387 sau .486.

Procedurile de mai sus operează cu numere în simplă precizie. Trecere la precizie dublă și extinsă se face însă foarte ușor. Trebuie doar ca variabilele externe și constantele reale utilizate să fie declarate de tip qword (8 octeți), respectiv tbytes (10 octeți), deoarece formatul intern al procesorului este cel de precizie maximă.

Deși procedurile care implică operații cu numere rele par complicate, implementările din acest subcapitol arată că o abordare sistematică și disciplinată permite dezvoltarea unui cod sursă clar, ușor de urmărit și ușor de întreținut.